

meta trimfill — Nonparametric trim-and-fill analysis of publication bias

Description	Quick start	Menu	Syntax
Options	Remarks and examples	Stored results	Methods and formulas
References	Also see		

Description

`meta trimfill` performs the nonparametric “trim-and-fill” method to account for publication bias in meta-analysis. The command estimates the number of studies potentially missing from a meta-analysis because of publication bias, imputes these studies, and computes the overall effect-size estimate using the observed and imputed studies. It can also provide a funnel plot, in which omitted studies are imputed.

Quick start

Perform the trim-and-fill analysis of publication bias under the default setting specified in either `meta set` or `meta esize`

```
meta trimfill
```

As above, and request a funnel plot

```
meta trimfill, funnel
```

Specify that the number of missing studies be estimated using the rightmost-run estimator instead of the default linear estimator

```
meta trimfill, estimator(run)
```

Specify that the estimation of the overall effect size be based on the fixed-effects inverse-variance method during the iteration step and random-effects DerSimonian–Laird method during the pooling step of the trim-and-fill algorithm

```
meta trimfill, itermethod(fixed) poolmethod(dlaird)
```

Specify that studies in the right side of the funnel plot (with large effect sizes) be suppressed and need to be imputed

```
meta trimfill, right
```

Menu

Statistics > Meta-analysis

Syntax

```
meta trimfill [if] [in] [, options]
```

<i>options</i>	Description
Main	
<code>estimator(<i>estimator</i>)</code>	estimator for the number of missing studies; default is <code>linear</code>
<code>left</code>	impute studies on the left side of the funnel plot
<code>right</code>	impute studies on the right side of the funnel plot
<code>funnel[(<i>funnelopts</i>)]</code>	draw funnel plot
Options	
<code>level(#)</code>	set confidence level; default is as declared for meta-analysis
<code>eform_option</code>	report exponentiated results
<code>[no]metashow</code>	display or suppress meta settings in the output
<code>display_options</code>	control column formats
Iteration	
<code>random[(<i>remethod</i>)]</code>	random-effects meta-analysis to use for iteration and pooling steps
<code>common</code>	common-effect meta-analysis to use for iteration and pooling steps; implies inverse-variance method
<code>fixed</code>	fixed-effects meta-analysis to use for iteration and pooling steps; implies inverse-variance method
<code>itermethod(<i>method</i>)</code>	meta-analysis to use for iteration step
<code>poolmethod(<i>method</i>)</code>	meta-analysis to use for pooling step
<code>iterate(#)</code>	maximum number of iterations for the trim-and-fill algorithm; default is <code>iterate(100)</code>
<code>[no]log</code>	display an iteration log from the trim-and-fill algorithm
<hr/>	
<i>estimator</i>	Description
<code>linear</code>	linear estimator, L_0 ; the default
<code>run</code>	run estimator, R_0
<code>quadratic</code>	quadratic estimator, Q_0 (rarely used)
<hr/>	
<i>remethod</i>	Description
<code>reml</code>	restricted maximum likelihood; the default
<code>mle</code>	maximum likelihood
<code>ebayes</code>	empirical Bayes
<code>dlaird</code>	DerSimonian–Laird
<code>sjonkman</code>	Sidik–Jonkman
<code>hedges</code>	Hedges
<code>hschmidt</code>	Hunter–Schmidt

Options

Main

`estimator` (*estimator*) specifies the type of estimator for the number of missing studies. *estimator* is one of `linear`, `run`, or `quadratic`. The default is `estimator(linear)`.

`linear` specifies that the “linear” estimator, L_0 , be used to estimate the number of missing studies. This is the default estimator.

`run` specifies that the rightmost-run estimator, R_0 , be used to estimate the number of missing studies.

`quadratic` specifies that the “quadratic” estimator, Q_0 , be used to estimate the number of missing studies. This estimator is not recommended in the literature and provided for completeness.

Duval and Tweedie (2000a) found that the L_0 and R_0 estimators perform better in terms of mean squared error (MSE) than the Q_0 estimator, with L_0 having the smallest MSE in certain cases. They also found that R_0 tends to be conservative in some cases. Therefore, L_0 is chosen to be the default, but the authors recommend that all estimators be considered in practice. Also see *Estimating the number of missing studies in Methods and Formulas* for details about the estimators.

`left` and `right` specify the side of the funnel plot, where the missing studies are to be imputed. By default, the side is chosen based on the results of the traditional Egger test—if the estimated slope is positive, `left` is assumed; otherwise, `right` is assumed. Only one of `left` or `right` is allowed.

`left` assumes that the leftmost (smallest) effect sizes have been suppressed and specifies to impute them.

`right` assumes that the rightmost (largest) effect sizes have been suppressed and specifies to impute them.

`funnel` and `funnel` (*funnelopts*) specify to draw a funnel plot that includes the imputed studies.

funnelopts are any options as documented in [META] [meta funnelplot](#), except `random[()]`, `common[()]`, `fixed[()]`, `by()`, `[no]metashow`, and `addplot()`.

Options

`level(#)` specifies the confidence level, as a percentage, for confidence intervals. The default is as declared for the meta-analysis session; see *Declaring a confidence level for meta-analysis* in [META] [meta data](#). Also see option `level()` in [META] [meta set](#).

eform_option is one of `eform`, `eform(string)`, `or`, or `rr`. It reports exponentiated effect sizes and transforms their respective confidence intervals, whenever applicable. By default, the results are displayed in the metric declared with `meta set` or `meta esize` such as log odds-ratios and log risk-ratios. *eform_option* affects how results are displayed, not how they are estimated and stored.

`eform(string)` labels the exponentiated effect sizes as *string*; the other options use default labels. The default label is specific to the chosen effect size. For example, option `eform` uses `Odds Ratio` when used with log odds-ratios declared with `meta esize` or `Risk Ratio` when used with the declared log risk-ratios. Option `or` is a synonym for `eform` when log odds-ratio is declared, and option `rr` is a synonym for `eform` when log risk-ratio is declared. If option `eslabel(eslab)` is specified during declaration, then `eform` will use the `exp(eslab)` label or, if `eslab` is too long, the `exp(ES)` label.

`metashow` and `nometashow` display or suppress the meta setting information. By default, this information is displayed at the top of the output. You can also specify `nometashow` with `meta update` to suppress the meta setting output for the entire meta-analysis session.

`display_options`: `cformat(%,fmt)`; see [R] [Estimation options](#).

Iteration

Options `random()`, `common`, and `fixed`, when specified with `meta trimfill`, temporarily override the global model declared by `meta set` or `meta esize` during the computation. These options specify that the same method be used during both iteration and pooling steps. To specify different methods, use options `itermethod()` and `poolmethod()`. Options `random()`, `common`, and `fixed` may not be combined. If these options are omitted, the declared meta-analysis model is assumed; see [Declaring a meta-analysis model](#) in [META] [meta data](#). Also see [Meta-analysis models](#) in [META] [Intro](#).

`random` and `random(remethod)` specify that a random-effects model be assumed for meta-analysis during iteration and pooling steps of the trim-and-fill algorithm; see [Random-effects model](#) in [META] [Intro](#).

`remethod` specifies the type of estimator for the between-study variance τ^2 . `remethod` is one of `reml`, `mle`, `ebayes`, `dlaird`, `sjonkman`, `hedges`, or `hschmidt`. `random` is a synonym for `random(reml)`. See [Options](#) in [META] [meta esize](#) for more information.

`common` specifies that a common-effect model be assumed for meta-analysis during iteration and pooling steps of the trim-and-fill algorithm; see [Common-effect \(“fixed-effect”\) model](#) in [META] [Intro](#). It uses the inverse-variance estimation method; see [Meta-analysis estimation methods](#) in [META] [Intro](#). Also see the [discussion](#) in [META] [meta data](#) about common-effect versus fixed-effects models.

`fixed` specifies that a fixed-effects model be assumed for meta-analysis during iteration and pooling steps of the trim-and-fill algorithm; see [Fixed-effects model](#) in [META] [Intro](#). It uses the inverse-variance estimation method; see [Meta-analysis estimation methods](#) in [META] [Intro](#). Also see the [discussion](#) in [META] [meta data](#) about fixed-effects versus common-effect models.

`itermethod(method)` specifies the meta-analysis method to use during the iteration step of the trim-and-fill algorithm. The default is the method declared for meta-analysis; see [Declaring a meta-analysis model](#) in [META] [meta data](#). Also see [Trim-and-fill algorithm](#) in [Methods and formulas](#). This option may not be combined with `random()`, `common`, or `fixed`.

`method` is one of the random-effects meta-analysis methods, `remethod`; or a common-effect inverse-variance method, `common`; or a fixed-effects inverse-variance method, `fixed`; see [Options](#) in [META] [meta set](#) for details.

`poolmethod(method)` specifies the meta-analysis method to use during the pooling step of the trim-and-fill algorithm. The default is to use the method declared for meta-analysis; see [Declaring a meta-analysis model](#) in [META] [meta data](#). Also see [Trim-and-fill algorithm](#) in [Methods and formulas](#). This option may not be combined with `random()`, `common`, or `fixed`.

`method` is one of the random-effects meta-analysis methods, `remethod`; or a common-effect inverse-variance method, `common`; or a fixed-effects inverse-variance method, `fixed`; see [Options](#) in [META] [meta set](#) for details.

`iterate(#)` specifies the maximum number of iterations for the trim-and-fill algorithm. The default is `iterate(100)`. When the number of iterations equals `iterate()`, the algorithm stops and presents the current results. If convergence is not reached, a warning message is also displayed. If convergence is declared before this threshold is reached, the algorithm will stop when convergence is declared.

no`log` and `log` specify whether an iteration log showing the progress of the trim-and-fill algorithm is to be displayed. By default, the log is suppressed but you can specify `log` to display it.

Remarks and examples

[stata.com](https://www.stata.com)

Remarks are presented under the following headings:

Introduction

Using meta trimfill

Examples of using meta trimfill

Introduction

Publication bias is a serious problem in meta-analysis. It arises when the decision whether to publish a study depends on the statistical significance of the results of the study. Typically, more significant findings are more likely to get published. See *Publication bias* of [\[META\] Intro](#) for details.

Publication bias can be assessed visually with a funnel plot ([\[META\] meta funnelplot](#)). More formal tests for the presence of publication bias or, more generally, of small-study effects are also available ([\[META\] meta bias](#)). The focus of this entry is on assessing the impact of the publication bias on the results. One of the methods popular in practice is the so-called trim-and-fill method, introduced by [Duval and Tweedie \(2000a, 2000b\)](#).

The main goal of the trim-and-fill method is to evaluate the impact of publication bias on our final inference. The idea of the method is to iteratively estimate the number of studies potentially missing because of publication bias at the iteration stage. Then, at the final pooling stage, impute (fill in) the effect sizes and effect-size standard errors for these studies and use the completed set of studies to compute the overall effect-size estimate. For details, see *Trim-and-fill algorithm*.

Meta-analysis literature does not provide definitive recommendations for which model should be used during the iteration stage of the method. [Duval \(2005\)](#) points out that a random-effects model was recommended initially ([National Research Council 1992](#); [Sutton et al. 1998](#)), but a common-effect model was found later to provide a more conservative approach. What is meant by conservative? Random-effects models tend to give more weight to less precise studies than common-effect models ([Poole and Greenland 1999](#)). But less precise (smaller) studies are more likely to exhibit publication bias. In general, neither model outperforms the other in all situations. Thus, meta-analysts are advised to try both in practice and compare the results.

Just like other methods for detecting publication bias such as funnel plots and tests for the funnel-plot asymmetry, the trim-and-fill method is sensitive to the presence of substantial between-study heterogeneity (for example, [Peters et al. \[2007\]](#)). The method is agnostic to the reasons for the funnel-plot asymmetry, be it publication bias or between-study heterogeneity. It merely detects the asymmetry and attempts to correct it. Unlike the tests for the funnel-plot asymmetry ([\[META\] meta bias](#)), the trim-and-fill method does not allow accounting for the moderators that could potentially explain the heterogeneity. For categorical moderators, however, you can perform the method separately for each category. In any case, you should take potential heterogeneity into account when interpreting the final estimate of the effect size from the trim-and-fill method. See the *Trim and fill* section in [Deeks, Higgins, and Altman \(2017\)](#) for details.

[Duval \(2005\)](#) states that there are researchers who are not comfortable with using imputed (fictional) studies to adjust the meta-analysis results for publication bias ([Begg 1997](#)). Indeed, one of the shortcomings of this method is that it treats “imputed” effect sizes as if they were observed in the final meta-analysis and thus potentially underestimates their corresponding standard errors.

Note that the emphasis of the trim-and-fill method is on sensitivity analyses and not on trying to recover the missing study values. The actual imputed values are not of interest. The primary goal of the method is to explore the impact of missing studies on the overall effect-size estimate. This method should be used purely as a guide to which meta-analysis appears to be more susceptible to publication bias.

Using meta trimfill

`meta trimfill` implements the trim-and-fill method of [Duval and Tweedie \(2000a, 2000b\)](#). It supports three estimators: `linear`, `run`, and `quadratic`, which can be specified in the `estimator()` option.

It can impute studies on the left side of the funnel plot with option `left` or on the right with option `right`. By default, it assumes the side where the missingness occurs based on the traditional Egger test; see [\[META\] meta bias](#). Specifically, if the slope from the Egger test is positive, then `left` is assumed; otherwise, `right` is assumed.

You can use option `funnel` or `funnel()` to draw a funnel plot, which would include the imputed studies. You can customize the default look of the funnel plot by specifying `funnel()`.

You can use `eform_option` to exponentiate the results reported in the log metric such as log odds-ratios and log risk-ratios.

By default, the model declared with either `meta set` or `meta esize` is assumed during both the iteration and pooling stages. You can change this by specifying one of `random()`, `common`, or `fixed`. You can also specify a different model for the iteration stage in the `itermethod()` option and for the pooling stage in the `poolmethod()` option.

Examples of using meta trimfill

Here we use `metatrim.dta` to demonstrate the usages of the trim-and-fill method. (This dataset is based on `pupiliq.dta`; it contains a subset of variables renamed to have generic names and fewer observations.) Also see [example 13](#) of [\[META\] meta](#) for the trim-and-fill analysis of the [NSAIDS](#) dataset.

► Example 1: A basic trim-and-fill analysis

Consider the dataset `metatrim.dta`. In this dataset, each study consists of two groups, and the standardized difference in means between the two groups is recorded, as well as the standard error for this difference. We begin by declaring the effect sizes and their standard errors with `meta set`.

```

. use https://www.stata-press.com/data/r16/metatrim
(Substet of pupiliq.dta)
. meta set stdmdiff se
Meta-analysis setting information
Study information
  No. of studies: 16
  Study label: Generic
  Study size: N/A
  Effect size
    Type: Generic
    Label: Effect Size
    Variable: stdmdiff
  Precision
  Std. Err.: se
    CI: [_meta_cil, _meta_ciu]
  CI level: 95%
Model and method
  Model: Random-effects
  Method: REML

```

We perform a trim-and-fill analysis to see whether any studies are estimated to be missing:

```

. meta trimfill
  Effect-size label: Effect Size
  Effect size: stdmdiff
  Std. Err.: se
Nonparametric trim-and-fill analysis of publication bias
Linear estimator, imputing on the left
Iteration                                Number of studies = 19
  Model: Random-effects                    observed = 16
  Method: REML                             imputed = 3
Pooling
  Model: Random-effects
  Method: REML

```

Studies	Effect Size	[95% Conf. Interval]	
Observed	0.119	-0.018	0.256
Observed + Imputed	0.034	-0.150	0.219

The model in the iteration and pooling steps is a random-effects model with the REML estimation as declared by `meta set`. These models may be individually controlled by using options `itermethod()` and `poolmethod()`. By default, estimation of the number of missing studies K_0 was based on the linear estimator. This can be changed using the `estimator()` option.

The mean effect size based on the 16 observed studies is 0.119 with a 95% CI of $[-0.018, 0.256]$. Three hypothetical studies, $\widehat{K}_0 = 19 - 16 = 3$, are estimated to be missing and are imputed. If these three studies were included in the meta-analysis, the funnel plot would be more symmetrical. After imputing the studies, we obtain an updated estimate (based on the 19 studies, observed plus imputed) of the mean effect size of 0.034 with a 95% CI $[-0.150, 0.219]$.

Example 2: Funnel plot

Continuing with [example 1](#), we can request a funnel plot by specifying the `funnel` option.


```
. meta trimfill, funnel
      Effect-size label: Effect Size
      Effect size: stmdiff
      Std. Err.: se

Nonparametric trim-and-fill analysis of publication bias
Linear estimator, imputing on the left

Iteration                                Number of studies =    19
  Model: Random-effects                   observed =          16
  Method: REML                             imputed =           3

Pooling
  Model: Random-effects
  Method: REML
```


Studies	Effect Size	[95% Conf. Interval]	
Observed	0.119	-0.018	0.256
Observed + Imputed	0.034	-0.150	0.219



The imputed studies are shown in orange in the above funnel plot. Only one of the three imputed studies lies within the 95% pseudo CI; see [\[META\] meta funnelplot](#) for the interpretation of the funnel plot.

We may be interested in a contour-enhanced funnel plot of the completed set of studies (observed and imputed) to visually assess whether the imputed studies fall in regions of statistical significance. This is done by specifying the `contour(1 5 10)` suboption within `funnel()` in `meta trimfill`. We suppress the output from the command with `quietly`.


```
. quietly meta trimfill, funnel(contour(1 5 10))
```



The plot reveals that two of the three imputed studies fall in the white region corresponding to a p -value less than 1%; see [example 5](#) of [\[META\] meta funnelplot](#) for more examples of contour-enhanced funnel plots.

◀

► Example 3: Specifying different pooling and iteration methods

By default, the same meta-analysis methods will be used for both the pooling and iteration steps, but we can specify other methods with the `poolmethod()` and `itermethod()` options. For example, below we specify the random-effects DerSimonian–Laird method for the pooling step and the fixed-effects inverse-variance method for the iteration step:

```
. meta trimfill, itermethod(fixed) poolmethod(dlaird)
Effect-size label: Effect Size
Effect size: stdmdiff
Std. Err.: se

Nonparametric trim-and-fill analysis of publication bias
Linear estimator, imputing on the left

Iteration                                Number of studies =    19
Model: Fixed-effects                      observed =           16
Method: Inverse-variance                  imputed =            3

Pooling
Model: Random-effects
Method: DerSimonian-Laird
```

Studies	Effect Size	[95% Conf. Interval]	
Observed	0.117	-0.016	0.249
Observed + Imputed	0.033	-0.120	0.186

The estimates are only slightly smaller than what they were in [example 1](#), where we used the random-effects REML estimation method in both steps.

◀

▷ Example 4: Specifying the estimator

By default, the linear estimator is used to estimate the number of missing studies. Let's use the rightmost-run estimator and see how the results differ:

```
. meta trimfill, estimator(run)
      Effect-size label: Effect Size
      Effect size: stdmdiff
      Std. Err.: se

Nonparametric trim-and-fill analysis of publication bias
Run estimator, imputing on the left

Iteration                                Number of studies =    18
  Model: Random-effects                    observed =          16
  Method: REML                              imputed =           2

Pooling
  Model: Random-effects
  Method: REML
```

	Studies	Effect Size	[95% Conf. Interval]	
	Observed	0.119	-0.018	0.256
	Observed + Imputed	0.059	-0.124	0.242

The mean effect size based on the 16 observed studies is still 0.119, as it was in [example 1](#), but the updated estimate that includes the imputed studies is large now, 0.059. Also, the estimated number of missing studies is 2 in this example instead of 3.

◀

Stored results

`meta trimfill` stores the following in `r()`:

Scalars

`r(K_total)` total number of studies (observed plus imputed)
`r(K_observed)` number of observed studies
`r(K_imputed)` number of imputed studies
`r(converged)` 1 if trim-and-fill algorithm converged, 0 otherwise

Macros

`r(estimator)` type of estimator for the number of missing studies
`r(side)` side of the funnel plot with missing studies; `left` or `right`
`r(itermethod)` meta-analysis estimation method used during iteration step
`r(poolmethod)` meta-analysis estimation method used during final pooling step
`r(level)` confidence level for CIs

Matrices

`r(table)` trim-and-fill table of results
`r(imputed)` matrix of effect sizes and their standard errors for imputed studies

Methods and formulas

For each study j , $j = 1, \dots, K$, let $\hat{\theta}_j$ be the effect size and $\hat{\sigma}_j^2$ be its squared standard error (or within-study variance). The goal is to estimate an overall effect size, θ , from the sample of the effect sizes, accounting for (potentially) suppressed studies with extreme values from the meta-analysis.

The formulas and discussion below are based on Duval and Tweedie (2000b), Duval (2005), and Steichen (2000).

Suppose that there are K observed studies and K_0 relevant studies that are potentially missing from the meta-analysis because of publication bias. The goal is to estimate the value of K_0 as well as the effect size from the “completed” set of $K + K_0$ studies. The formulas below are based on the assumption that the K_0 effect-size values that are missing are the most extreme smallest values. That is, the studies with “nonsignificant” results are the ones being suppressed. This is also equivalent to assuming that the studies are missing from the left side of the funnel plot (option `left`). If missing studies are expected to be missing on the right side of the funnel (option `right`), Duval (2005) indicates that the same formulas below can be applied after multiplying the effect sizes by -1 .

Note that the default behavior of `meta trimfill` is to assume the side where the missingness occurs based on the traditional Egger test; see *Egger’s linear regression test* in [META] **meta bias**. Specifically, if the slope from the Egger test is positive, then option `left` is assumed, and vice versa.

Methods and formulas are presented under the following headings:

Estimating the number of missing studies
Trim-and-fill algorithm

Estimating the number of missing studies

We follow the description and notation from Duval and Tweedie (2000b) and Steichen (2000). Let $X_j = \hat{\theta}_j - \hat{\theta}$ and denote the ranks of $|X_j|$ ’s as $r_j^* \in \{1, 2, \dots, K\}$. Let $\gamma^* \geq 0$ be the length of the rightmost run of ranks associated with positive values of X_j ’s such that $\gamma^* = K - r_h^*$, where r_h^* is the absolute rank of the most negative X_j with the corresponding index h ; see Rothstein et al. (2005, 137) for an example illustrating the computation of γ^* manually.

The “trimmed” rank test statistic for the observed K values is

$$T_K = \sum_{X_j > 0} r_j^*$$

The following estimators of K_0 can be considered based on the above quantities:

$$\begin{aligned} R_0 &= \gamma^* - 1 \\ L_0 &= \frac{4T_K - K(K + 1)}{2K - 1} \\ Q_0 &= K - \frac{1}{2} - \sqrt{2K^2 - 4T_K + \frac{1}{4}} \end{aligned}$$

Because K_0 must be an integer, the above estimators are rounded as follows,

$$\begin{aligned} R_0^+ &= \max(0, R_0) \\ L_0^+ &= \max\{0, \text{round}(L_0)\} \\ Q_0^+ &= \max\{0, \text{round}(Q_0)\} \end{aligned}$$

where $\text{round}(x)$ is x rounded to the nearest integer.

Duval and Tweedie (2000a) compared the behavior of the MSEs of the estimators R_0^+ , L_0^+ , and Q_0^+ using simulated data and found that L_0^+ and R_0^+ seem to perform better than Q_0^+ . They also found that R_0 tends to be conservative in some cases. In general, the authors recommend that all estimators be considered in practice. Also see Duval and Tweedie (2000b), Duval (2005), and Steichen (2016) for more details about these estimators.

Trim-and-fill algorithm

Without loss of generality, we assume that the observations, $\hat{\theta}_j$'s, are sorted in ascending order.

The steps of the trim-and-fill method below are based on Duval and Tweedie (2000b) and Steichen (2016).

Step 1. At the initial iteration $l = 1$, compute $\hat{\theta}^{(1)}$ using the meta-analysis model as declared with `meta set` or `meta esize`; the model may be changed using the `itermethod()` option.

Compute the centered values

$$\hat{\theta}_j^{(1)} = \hat{\theta}_j - \hat{\theta}^{(1)} \quad j = 1, \dots, K$$

and estimate $\hat{K}_0^{(1)}$ using the default L_0^+ estimator applied to the set of centered values $\hat{\theta}_j^{(1)}$; that is, set $X_j = \hat{\theta}_j^{(1)}$ in the previous section. Other estimators for $\hat{K}_0^{(1)}$ may be specified using the `estimator()` option.

Step 2. At the iteration $l \geq 2$, remove $\hat{K}_0^{(l-1)}$ values from the right end of the set of values $\hat{\theta}_j$, and estimate $\hat{\theta}^{(l)}$ based on the trimmed “symmetric” set of $K - \hat{K}_0^{(l-1)}$ values: $\{\hat{\theta}_1, \dots, \hat{\theta}_{K-\hat{K}_0^{(l-1)}}\}$. Compute the next set of centered values

$$\hat{\theta}_j^{(l)} = \hat{\theta}_j - \hat{\theta}^{(l)} \quad j = 1, \dots, K$$

Step 3. Repeat step 2 until an iteration $l = L$, at which $\hat{K}_0^{(L)} = \hat{K}_0^{(L-1)}$ (and thus, $\hat{\theta}^{(L)} = \hat{\theta}^{(L-1)}$). Set $\hat{K}_0 = \hat{K}_0^{(L)}$.

Step 4. Finally, compute the \hat{K}_0 imputed symmetric values

$$\hat{\theta}_j^* = 2\hat{\theta}^{(L)} - \hat{\theta}_{K-j+1} \quad j = 1, \dots, \hat{K}_0$$

and the corresponding imputed within-study standard errors,

$$\hat{\sigma}_j^* = \hat{\sigma}_{K-j+1} \quad j = 1, \dots, \hat{K}_0$$

Compute the final overall effect-size estimate using the default meta-analysis method or `poolmethod()`, if specified, based on the “completed” dataset $\{\hat{\theta}_1, \dots, \hat{\theta}_K, \hat{\theta}_1^*, \dots, \hat{\theta}_{\hat{K}_0}^*\}$.

References

- Begg, C. B. 1997. Publication bias in meta-analysis: A Bayesian data-augmentation approach to account for issues exemplified in the passive smoking debate: Comment. *Statistical Science* 12: 241–244.
- Deeks, J. J., J. P. T. Higgins, and D. G. Altman. 2017. Addressing reporting biases. In *Cochrane Handbook for Systematic Reviews of Interventions Version 5.2.0*, ed. J. A. C. Sterne, M. Egger, and D. Moher, chap. 10. London: The Cochrane Collaboration. <https://training.cochrane.org/handbook>.
- Duval, S. 2005. The trim and fill method. In *Publication Bias in Meta-Analysis: Prevention, Assessment and Adjustments*, ed. H. R. Rothstein, A. J. Sutton, and M. Borenstein, 127–144. Chichester, UK: Wiley.
- Duval, S., and R. L. Tweedie. 2000a. Trim and fill: A simple funnel-plot-based method of testing and adjusting for publication bias in meta-analysis. *Biometrics* 56: 455–463.
- . 2000b. A nonparametric “trim and fill” method of accounting for publication bias in meta-analysis. *Journal of American Statistical Association* 95: 89–98.
- National Research Council. 1992. *Combining Information: Statistical Issues and Opportunities for Research*. Washington, DC: National Academies Press.
- Peters, J. L., A. J. Sutton, D. R. Jones, K. R. Abrams, and L. Rushton. 2007. Performance of the trim and fill method in the presence of publication bias and between-study heterogeneity. *Statistics in Medicine* 26: 4544–4562.
- Poole, C., and S. Greenland. 1999. Random-effects meta-analyses are not always conservative. *American Journal of Epidemiology* 150: 469–475.
- Rothstein, H. R., A. J. Sutton, and M. Borenstein, ed. 2005. *Publication Bias in Meta-Analysis: Prevention, Assessment and Adjustments*. Chichester, UK: Wiley.
- Steichen, T. J. 2000. [sbe39: Nonparametric trim and fill analysis of publication bias in meta-analysis](#). *Stata Technical Bulletin* 57: 8–14. Reprinted in *Stata Technical Bulletin Reprints*, vol. 10, pp. 108–117. College Station, TX: Stata Press.
- . 2016. Nonparametric trim and fill analysis of publication bias in meta-analysis. In *Meta-Analysis in Stata: An Updated Collection from the Stata Journal*, ed. T. M. Palmer and J. A. C. Sterne, 2nd ed., 180–192. College Station, TX: Stata Press.
- Sutton, A. J., K. R. Abrams, D. R. Jones, T. A. Sheldon, and F. Song. 1998. Systematic reviews of trials and other studies. *Health Technology Assessment* 2(19): 1–276.

Also see

- [META] [meta bias](#) — Tests for small-study effects in meta-analysis
- [META] [meta data](#) — Declare meta-analysis data
- [META] [meta funnelplot](#) — Funnel plots
- [META] [meta summarize](#) — Summarize meta-analysis data
- [META] [meta](#) — Introduction to meta
- [META] [Glossary](#)
- [META] [Intro](#) — Introduction to meta-analysis